

HEWLETT-PACKARD

Budgeting & Finance I

Business Software Solutions

SERIES 80

17
18
190
200
210 P
220 MOVE
230 FOR R
240 D=P*(N-
*FNF(N-K
250 PRINT R, 1, 0*1000+.5)/1000
260 DRAW R,D
270 IDRAW 1,0
280 NEXT R
290 DRAW 11,0
300 DEF FNF(X)
310 F=1
320 FOR I=X TO 1 STEP -1
330 F=F*I
340 NEXT I
350 FNF=F
360 FN END

NO. 2 NET WEIGHT DISTRIBUTION

HP-85

EXAM GRADES

LIM: 48 E 43 D 35 C 22 B 30 33 H

TABLE OF CONTENTS

1.	PROFORMA INCOME STATEMENT AND BALANCE SHEET	1
	Provides a listing of a simple proforma income statement and balance sheet.	
2.	SALES COMMISSION REPORT	11
	Computes a monthly planning table for a salesperson, with their base salary, incentive dollars, and prospective billing totals.	
3.	FINANCIAL RATIOS	18
	Uses income statement and balance sheet data to compute various financial ratios.	
4.	LEASE INCOME	24
	Calculates annual lease income from (U) units at sales price (S) leased at lease rate (R) for lease period (L). It also sums the total income by year over (Y) years of lease operation.	
5.	MAKE-BUY DECISION ANALYSIS	30
	Calculates the present value of the cost saving incurred by making a product as opposed to buying it. It also prints a cash flow summary for each method for each year involved.	
6.	LOAN AMORTIZATION	39
	Amortizes a loan using arbitrary monthly payments entered by the user. Monthly reports are generated indicating payments to interest and principal. Yearly reports are printed showing accumulated payments to interest and principal and the remaining balance. Final payment is computed when balance falls below user-set maximum monthly payment.	
7.	MORTGAGE ANALYSIS	46
	Will find the missing parameter of the following four, given the remaining three: Interest rate, life (years), amount borrowed, and monthly payment. The program will then print a monthly or yearly summary indicating the amount of interest, amount of payment and outstanding balance for each period.	
8.	COST OF EQUITY CAPITAL	53
	Computes the cost of equity capital by computing the dividends and the share price for future periods, and finds the discount rate by equating the present value of the stream to the current share price.	
9.	DEPRECIATION METHOD COMPARISON	59
	Computes and prints the monthly depreciation of a given investment by four methods: straight line, double declining balance, sum of the years digits, and 150% declining balance.	
10.	SIMPLE LOAN ANALYSIS	68
	Calculates the monthly interest charges and outstanding balance of a loan that must be paid off in one year or less.	

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION I

(Please Type)

Program Title Proforma Income Statement and Balance Sheet

File Name I N C O M E

Contributor's Name

Company (if applicable)

Address

City

State/Country

Zip Code/Mail Code

Machine Size: 16K ☒ 32K ☐

Peripherals Required: none

ROMs Required: none

Number of Bytes: 6,937

Program Description: INCOME provides a listing of a simple proforma income statement and balance sheet.

User accepts and uses this program material AT HIS/HER OWN RISK, in reliance solely upon his/her own inspection of the program material and without reliance upon any representation or description concerning the program material.
NEITHER HP NOR THE CONTRIBUTOR MAKES ANY EXPRESS OR IMPLIED WARRANTY OF ANY KIND WITH REGARD TO THIS PROGRAM MATERIAL, INCLUDING, BUT NOT LIMITED TO THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE NEITHER HP NOR THE CONTRIBUTOR SHALL BE LIABLE FOR INCIDENTAL OR CONSEQUENTIAL DAMAGES IN CONNECTION WITH OR ARISING OUT OF THE FURNISHING, USE OR PERFORMANCE OF THIS PROGRAM MATERIAL.

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II

Sample Problem (Sketch if Desired)

Generate income statement using following input data:

Base period sales total	:	50000
Net fixed assets	:	450000
Remaining long term debt	:	35000
Total owner's equity	:	800000
Retained earnings	:	31000
Federal corporate tax rate	:	.48
Projected quarterly dividends:		80000
Quarterly retained cash	:	300000
Accounts receivable turnover	:	2.25
Inventory turnover	:	1.5
Accounts payable turnover	:	.20

	<u>Qtr 1</u>	<u>Qtr 2</u>	<u>Qtr 3</u>	<u>Qtr 4</u>
Est. growth percentage sales :	.1	.1	.15	.2
Cost of goods sold as % sales:	.48	.49	.5	.51
G S & A expense as % of sales:	.1	.12	.12	.12
Fixed asset purchases :	2000	5000	5000	1000
Fixed asset retirements :	1000	1000	1000	1000
Planned debt payments :	2500	2500	2500	2500

SOLUTION:

LOAD "INCOME"
RUN

Enter data as requested.

*** INCOME STATEMENT ***

QUARTER NO.	1
SALES	55000.00
COST OF GOODS SOLD	26400.00
GS AND A EXPENSES	5500.00
PROFIT BEFORE TAX	23100.00
FED. INCOME TAX	11088.00
PROFIT AFTER TAX	12012.00
ANNUAL DIVIDENDS	8000.00

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II

QUARTER NO. 2

SALES	60500.00
COST OF GOODS SOLD	29645.00
GS AND A EXPENSES	7260.00
PROFIT BEFORE TAX	23595.00
FED. INCOME TAX	11325.60
PROFIT AFTER TAX	12269.40
ANNUAL DIVIDENDS	8000.00

QUARTER NO. 3

SALES	69575.00
COST OF GOODS SOLD	34787.50
GS AND A EXPENSES	8349.00
PROFIT BEFORE TAX	26438.50
FED. INCOME TAX	12690.48
PROFIT AFTER TAX	13748.02
ANNUAL DIVIDENDS	8000.00

QUARTER NO. 4

SALES	83490.00
COST OF GOODS SOLD	42579.90
GS AND A EXPENSES	10018.80
PROFIT BEFORE TAX	30891.30
FED. INCOME TAX	14827.82
PROFIT AFTER TAX	16063.48
ANNUAL DIVIDENDS	8000.00

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II

*** BALANCE SHEET ***

QUARTER NO. 3

QUARTER NO. 1

CASH	377488.89
ACCOUNTS RECEIVABLE	12012.00
INVENTORY	36666.67
TOTAL CURRENT ASSETS	438600.00
NET FIXED ASSETS	451000.00
TOTAL ASSETS	889600.00
ACCOUNTS PAYABLE	11000.00
ACCRUED TAXES	11088.00
TOTAL CURRENT LIAB.	22088.00
DEBT FINANCING	32500.00
EQUITY	800000.00
RETAINED EARNINGS	35012.00
TOTAL LIABILITIES	889600.00

QUARTER NO. 2

CASH	370484.78
ACCOUNTS RECEIVABLE	12269.40
INVENTORY	40333.33
TOTAL CURRENT ASSETS	437707.00
NET FIXED ASSETS	455000.00
TOTAL ASSETS	892707.00
ACCOUNTS PAYABLE	12100.00
ACCRUED TAXES	11325.60
TOTAL CURRENT LIAB.	23425.60
DEBT FINANCING	30000.00
EQUITY	800000.00
RETAINED EARNINGS	39281.40
TOTAL LIABILITIES	892707.00

CASH	362829.34
ACCOUNTS RECEIVABLE	13748.02
INVENTORY	46383.33
TOTAL CURRENT ASSETS	440134.90
NET FIXED ASSETS	459000.00
TOTAL ASSETS	899134.90
ACCOUNTS PAYABLE	13915.00
ACCRUED TAXES	12690.48
TOTAL CURRENT LIAB.	26605.48
DEBT FINANCING	27500.00
EQUITY	800000.00
RETAINED EARNINGS	45029.42
TOTAL LIABILITIES	899134.90

QUARTER NO. 4

CASH	357852.05
ACCOUNTS RECEIVABLE	16063.48
INVENTORY	55660.00
TOTAL CURRENT ASSETS	450618.72
NET FIXED ASSETS	459000.00
TOTAL ASSETS	909618.72
ACCOUNTS PAYABLE	16698.00
ACCRUED TAXES	14827.82
TOTAL CURRENT LIAB.	31525.82
DEBT FINANCING	25000.00
EQUITY	800000.00
RETAINED EARNINGS	53092.90
TOTAL LIABILITIES	909618.72

SERIES 80 USERS' LIBRARY
PROGRAM DESCRIPTION III

Operating Limits and Warnings Enter all percentages as a decimal less than 1.

Reference(s)

Variables:

Name	Description	Length	Comments
B1	Base period sales total		
B2	Base period net fixed assets		
B4	Base period long-term debt remaining		
B5	Base period total owners' equity		
B6	Base period accumulated retained earnings		
S2	Federal corporate tax rate on profit		
S3	Promised period dividends		
R1	Amount of cash the user wishes to hold after base period		
R2	Accounts receivable turnover		
R3	Inventory turnover		
R6	Accounts Payable turnover		
L(1,1-4)	Estimated percentage of growth in sales for next 4 months		
L(2,1-4)	Cost of goods sold as an estimated percentage of sales		
L(3,1-4)	G, S & A expenses as an estimated percentage of sales		
L(4,1-4)	Estimated amount of fixed assets purchases		
L(5,1-4)	Estimated amount of fixed assets retired		
L(6,1-4)	Planned payments per quarter on debt		

SERIES 80 USERS' LIBRARY
USER INSTRUCTIONS

STEP**INSTRUCTIONS**

- 1 LOAD "INCOME" and press RUN.
- 2 Enter the input data as requested.
 Be sure to enter percentages as decimals less than 1.
- 3 Income statement and Balance sheet will be printed.

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```

10 CLEAR
20 OPTION BASE 1
30 DIM L(6,4)
40 DISP "WHAT IS THE BASE PERIO
D'S TOTAL SALES";@ INPUT B1
50 DISP @ DISP "WHAT ARE THE NE
T FIXED ASSETS FOR THE BAS
E PERIOD";@ INPUT B2
60 DISP @ DISP "WHAT IS THE REM
AINING LONG TERM DEBT FOR TH
E BASE PERIOD";@ INPUT B4
70 DISP @ DISP "WHAT IS THE TOT
AL OWNER'S EQUITYFOR THE BAS
E PERIOD";@ INPUT B5
80 DISP @ DISP "WHAT ARE THE RE
TAINED EARNINGS FOR THE BAS
E PERIOD";@ INPUT B6
90 DISP @ DISP "WHAT IS THE FED
ERAL CORPORATE TAX RATE";@
INPUT S2
100 IF S2<1 THEN 120
110 GOSUB 1490 @ GOTO 90
120 DISP @ DISP "WHAT IS THE QUA
RTERLY PROJECTED AMOUNT OF D
IVIDENDS";@ INPUT S3
130 DISP @ DISP "WHAT AMOUNT OF
CASH WOULD YOU LIKE TO RET
AIN FOR EACH QUARTER" @ INPU
T R1
140 DISP @ DISP "WHAT IS THE NOR
MAL ACCT'S/RECV TURNOVER";@
INPUT R2
150 IF R2>1 THEN 170
160 GOSUB 1500 @ GOTO 140
170 DISP @ DISP "WHAT IS THE NOR
MAL INVENTORY TURNOVER";@
INPUT R3
180 IF R3>1 THEN 200
190 GOSUB 1500 @ GOTO 170
200 DISP @ DISP "WHAT PERCENTAGE
OF SALES ARE THEACCOUNTS/PA
YABLE (A/P TURNOVER)";@ INPU
T R6
210 IF R6<1 THEN 230
220 GOSUB 1490 @ GOTO 200
230 R8=0
240 CLEAR
250 DISP "ENTER FOUR VALUES FOR
EACH QUES-TION. EACH VALUE
APPLIES TO THERESPECTIVE QUA
RTER:"
260 DISP @ DISP "ENTER THE ESTIM
ATED PERCENTAGE GROWTH IN S
ALES";
270 INPUT L(1,1),L(1,2),L(1,3),L
(1,4)
280 IF L(1,1)<1 THEN 300
290 GOSUB 1490 @ GOTO 260
300 DISP @ DISP "ENTER THE COST
OF GOODS SOLD AS ESTIMATED P
ERCENTAGE OF SALES";

```

Data entry

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```

310 INPUT L(2,1),L(2,2),L(2,3),L
    (2,4)
320 IF L(2,1)<1 THEN 340
330 GOSUB 1490 @ GOTO 300
340 DISP @ DISP "ENTER THE GENER
    AL SELLING AND ADMINISTRAT
    IVE EXPENSE AS A PER-"
350 DISP "CENTAGE OF SALES";@ IN
    PUT L(3,1),L(3,2),L(3,3),L(3
    ,4)
360 IF L(3,1)<1 THEN 380
370 GOSUB 1490 @ GOTO 350
380 DISP @ DISP "ENTER THE ESTIM
    ATED AMOUNT OF FIXED ASSET
    PURCHASES FOR EACH QUARTER
    ";
390 INPUT L(4,1),L(4,2),L(4,3),L
    (4,4)
400 DISP @ DISP "ENTER THE ESTIM
    ATED AMOUNT OF FIXED ASSET
    RETIREMENTS FOR EACHQUARTER
    ";
410 INPUT L(5,1),L(5,2),L(5,3),L
    (5,4)
420 DISP @ DISP "ENTER THE PLANN
    ED DEBT PAYMENTS FOR EACH QU
    ARTER";
430 INPUT L(6,1),L(6,2),L(6,3),L
    (6,4)
440 CLEAR @ DISP @ DISP @ DISP @
    DISP " PRINT OPTIONS:"
450 DISP @ DISP " 1) BALANCE
    SHEET ONLY"
460 DISP @ DISP " 2) INCOME
    STATEMENT ONLY"
470 DISP @ DISP " 3) BOTH ST
    ATEMENTS"
480 DISP @ DISP " SELECT PRI
    NT OPTION";@ INPUT K6
490 CLEAR
500 !
510 !
520 B7=B4 @ B8=B5 @ R8=1
530 FOR X=1 TO 4
540 A(X)=S3 @ A9=B1*L(1,X)
550 B1=B1+A9 @ B(X)=B1
560 C(X)=B(X)*L(2,X)
570 X(X)=B(X)*L(3,X)
580 K(X)=B(X)-(C(X)+X(X))
590 D(X)=K(X)*S2
600 U(X)=K(X)-D(X)
610 E(X)=K(X)-D(X)-S3
620 G(X)=B(X)/R2
630 H(X)=B(X)/R3
640 B2=B2+(L(4,X)-L(5,X))
650 N(X)=B2
660 O(X)=R1+N(X)+G(X)+H(X)
670 I(X)=B(X)*R6
680 J(X)=I(X)+D(X)
690 B4=B4-L(6,X)

```

Data entry

Select print option

Begin computation

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```

700 B6=B6+E(X)
710 S(X)=J(X)+B4+B5+B6
720 T(X)=O(X)-S(X)
730 IF T(X)>0 THEN 760
740 Z(X)=R1-T(X)
750 GOTO 840
760 D1=R8*(B5+B6)
770 Z(X)=R1
780 D2=D1-(B4+T(X))
790 IF D2<0 THEN 820
800 B4=B4+T(X)
810 GOTO 840
820 B5=B5-D2
830 B4=B4+(T(X)+D2)
840 P(X)=B4
850 M(X)=B5
860 R(X)=B6
870 F(X)=G(X)+H(X)+Z(X)
880 S(X)=J(X)+P(X)+M(X)+R(X)
890 O(X)=N(X)+G(X)+H(X)+Z(X)
900 NEXT X
910 Y(1)=P(1)-B7
920 W(1)=M(1)-R8
930 FOR X=2 TO 4
940 Y(X)=P(X)-P(X-1)
950 W(X)=M(X)-M(X-1)
960 NEXT X
970 !
980 !
990 IF K6=1 THEN 1180
1000 PRINT "    *** INCOME STATE
      MENT ***"
1010 PRINT "    -----
      -----"
1020 PRINT @ PRINT
1030 FOR X=1 TO 4
1040 PRINT "          QUARTER NO.
      ";X
1050 PRINT "    -----
      --"
1060 IMAGE 20A,2X,7D,DD
1070 PRINT USING 1060 ; "SALES",
      B(X)
1080 PRINT USING 1060 ; "COST OF
      GOODS SOLD",C(X)
1090 PRINT USING 1060 ; "GS AND
      A EXPENSES",X(X)
1100 PRINT USING 1060 ; "PROFIT
      BEFORE TAX",K(X)
1110 PRINT
1120 PRINT USING 1060 ; "FED. IN
      COME TAX",D(X)
1130 PRINT USING 1060 ; "PROFIT
      AFTER TAX",U(X)
1140 PRINT USING 1060 ; "ANNUAL
      DIVIDENDS";A(X) @ PRINT @ P
      RINT @ NEXT X
1150 PRINT
1160 PRINT "*****
      *****"

```

Print income statement

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```

1170 PRINT
1180 IF K6=2 THEN 1480
1190 PRINT @ PRINT
1200 PRINT "    ***    BALANCE SHE
ET    ***"
1210 PRINT "    -----
-----"
1220 PRINT @ PRINT
1230 FOR X=1 TO 4
1240 PRINT "            QUARTER NO.
";X
1250 PRINT "    -----
--"
1260 PRINT
1270 PRINT USING 1060 ; "CASH",Z
(X)
1280 PRINT USING 1060 ; "ACCOUNT
S RECEIVABLE",U(X)
1290 PRINT USING 1060 ; "INVENTO
RY",H(X)
1300 PRINT USING 1060 ; "TOTAL C
URRENT ASSETS",F(X)
1310 PRINT
1320 PRINT USING 1060 ; "NET FIX
ED ASSETS";N(X)
1330 PRINT
1340 PRINT USING 1060 ; "TOTAL A
SSETS",O(X)
1350 PRINT
1360 PRINT USING 1060 ; "ACCOUNT
S PAYABLE",I(X)
1370 PRINT USING 1060 ; "ACCRUED
TAXES",D(X)
1380 PRINT USING 1060 ; "TOTAL C
URRENT LIAB.",J(X)
1390 PRINT
1400 PRINT USING 1060 ; "DEBT FI
NANCING",P(X)
1410 PRINT USING 1060 ; "EQUITY"
,M(X)
1420 PRINT USING 1060 ; "RETAIN
E EARNINGS",R(X)
1430 PRINT
1440 PRINT USING 1060 ; "TOTAL L
IABILITIES",S(X)
1450 PRINT @ PRINT
1460 NEXT X
1470 PRINT "*****
*****"
1480 END
1490 BEEP @ DISP @ DISP "Enter d
ecimal < 1" @ RETURN
1500 BEEP @ DISP @ DISP "Enter d
ecimal > 1" @ RETURN

```

Print balance sheet

Percentage error routine

Entry error routine

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION I

(Please Type)

Program Title Sales Commission Report

File Name S A L E S

Contributor's Name

Company (if applicable)

Address

City

State/Country

Zip Code/Mail Code

Machine Size: 16K ☒ 32K ☐

Peripherals Required: none

ROMs Required: none

Number of Bytes: 1,653

Program Description: SALES will print a monthly planning table for a salesperson, with their base salary, incentive dollars, and prospective billing totals.

User accepts and uses this program material AT HIS/HER OWN RISK, in reliance solely upon his/her own inspection of the program material and without reliance upon any representation or description concerning the program material.
NEITHER HP NOR THE CONTRIBUTOR MAKES ANY EXPRESS OR IMPLIED WARRANTY OF ANY KIND WITH REGARD TO THIS PROGRAM MATERIAL, INCLUDING, BUT NOT LIMITED TO THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE NEITHER HP NOR THE CONTRIBUTOR SHALL BE LIABLE FOR INCIDENTAL OR CONSEQUENTIAL DAMAGES IN CONNECTION WITH OR ARISING OUT OF THE FURNISHING, USE OR PERFORMANCE OF THIS PROGRAM MATERIAL.

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II

Sample Problem (Sketch if Desired)

Print a monthly planning table for a salesperson with the following input data:

```

Base annual salary      : 5000
Base commission percentage: .20
Base monthly quota      : 10
Expected # new accts/mo : 1
Expected initial revenue : 100
Expected rate of growth  : .10
Beginning mo. of analysis : 1
  
```

SOLUTION:

```

LOAD "SALES"
RUN
  
```

Enter input data as requested.

```

BASE SALARY      = 5000
MONTHLY QUOTA    = 10
INITIAL REVENUE  = 100
AVERAGE GROWTH  = 10 % / MONTH
  
```

```

-----
MONTH:  1
  BASE      : 416.67
  INCENTIVE  : 18.00
  TOTAL     : 434.67
  MONTHLY BILLING : 100.00
  TOTAL BILLING  : 100.00
  
```

```

-----
MONTH:  2
  BASE      : 416.67
  INCENTIVE  : 40.00
  TOTAL     : 456.67
  MONTHLY BILLING : 210.00
  TOTAL BILLING  : 310.00
  
```

```

-----
MONTH:  3
  BASE      : 416.67
  INCENTIVE  : 64.20
  TOTAL     : 480.87
  MONTHLY BILLING : 331.00
  TOTAL BILLING  : 641.00
  
```

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II

MONTH: 4
 BASE : 416.67
 INCENTIVE : 90.82
 TOTAL : 507.49
 MONTHLY BILLING : 464.10
 TOTAL BILLING : 1105.10

MONTH: 5
 BASE : 416.67
 INCENTIVE : 120.10
 TOTAL : 536.77
 MONTHLY BILLING : 610.51
 TOTAL BILLING : 1715.61

MONTH: 6
 BASE : 416.67
 INCENTIVE : 152.31
 TOTAL : 568.98
 MONTHLY BILLING : 771.56
 TOTAL BILLING : 2487.17

MONTH: 7
 BASE : 416.67
 INCENTIVE : 187.74
 TOTAL : 604.41
 MONTHLY BILLING : 948.72
 TOTAL BILLING : 3435.89

MONTH: 8
 BASE : 416.67
 INCENTIVE : 226.72
 TOTAL : 643.38
 MONTHLY BILLING : 1143.59
 TOTAL BILLING : 4579.48

MONTH: 9
 BASE : 416.67
 INCENTIVE : 269.59
 TOTAL : 686.26
 MONTHLY BILLING : 1357.95
 TOTAL BILLING : 5937.42

MONTH: 10
 BASE : 416.67
 INCENTIVE : 316.75
 TOTAL : 733.42
 MONTHLY BILLING : 1593.74
 TOTAL BILLING : 7531.17

MONTH: 11
 BASE : 416.67
 INCENTIVE : 368.62
 TOTAL : 785.29
 MONTHLY BILLING : 1853.12
 TOTAL BILLING : 9384.28

MONTH: 12
 BASE : 416.67
 INCENTIVE : 425.69
 TOTAL : 842.35
 MONTHLY BILLING : 2138.43
 TOTAL BILLING : 11522.71

YEARLY TOTALS:
 PAYROLL:
 BASE : 5000.00
 INCENTIVE : 2280.54
 TOTAL : 7280.54

MONTHLY REVENUE AT END OF
 CALENDAR YEAR : 2138.43

TOTAL BILLINGS : 11522.71

TOTAL PAYROLL AS A PERCENTAGE
 OF SALES : 63.18

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION III

Operating Limits and Warnings

Reference(s)

Variables:

Name	Description	Length	Comments
B	Base salary		
R	Base commission percentage		
Q	Base monthly quota		
N	Expected number of new accounts per month		
X	Expected initial revenue		
P	Expected rate of growth		
W1	Beginning month of analysis		
Q\$	Print option		
B1	Monthly base salary		
I1	Incentive		
T1	Total \$		
A4	Monthly billing		
A9	Total billing		
T9	Total earnings		
I3	Total incentive \$		

SERIES 80 USERS' LIBRARY
USER INSTRUCTIONS

STEP**INSTRUCTIONS**

- 1 LOAD "SALES" and press RUN.
- 2 Enter input data as requested.
- 3 Planning table will be printed out on an annual or a monthly basis, as requested.

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

<pre> 10 CLEAR 20 DIM Q\$(32) 30 A4=0 @ A9=0 @ T4=0 @ T9=0 @ I3=0 40 DISP "ENTER BASE ANNUAL SALA RY"; 50 INPUT B 60 DISP @ DISP "ENTER COMMISSIO N %"; 70 INPUT R 80 DISP @ DISP "ENTER MONTHLY Q UOTA"; 90 INPUT Q 100 B1=B/12 110 IF R>1 THEN R=R/100 120 DISP @ DISP "ENTER EXPECTED NUMBER OF NEW ACCOUNTS PE R MONTH"; 130 INPUT N 140 DISP @ DISP "ENTER EXPECTED INITIAL REVENUE AND RATE OF GROWTH"; 150 INPUT X,P 160 IF P>1 THEN P=P/100 170 DISP @ DISP "ENTER THE BEGIN NING MONTH OF THEANALYSIS (1 -12)"; 180 INPUT W1 190 DISP @ DISP "DO YOU WANT AN ANNUAL [A] OR MONTHLY [M] PRINTOUT"; 200 INPUT Q\$ 210 PRINT "BASE SALARY = ";B 220 PRINT "MONTHLY QUOTA = ";Q 230 PRINT "INITIAL REVENUE = ";X 240 PRINT "AVERAGE GROWTH = ";P *100; "% / MONTH" 250 PRINT @ PRINT 260 PRINT "-----" -----" @ PRINT 270 IMAGE 3X,16A," ",6D.DD 280 FOR Z=W1 TO 12 290 A4=N*X+A4*(1+P) 300 I1=(A4-Q)*R 310 IF I1<0 THEN I1=0 320 I3=I3+I1 330 T1=B1+I1 340 T9=T9+T1 350 A9=A9+A4 360 IF Q\$="A" THEN 440 370 PRINT "MONTH: ";Z 380 PRINT USING 270 ; "BASE",B1 390 PRINT USING 270 ; "INCENTIVE ",I1 400 PRINT USING 270 ; "TOTAL",T1 410 PRINT USING 270 ; "MONTHLY B ILLING",A4 420 PRINT USING 270 ; "TOTAL BIL LING",A9 430 PRINT @ PRINT " -----" -----" @ PRINT </pre>	<pre> Initialize Data entry Print initial conditions Begin computation Print monthly data </pre>
---------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------	------------------------------------------------------------------------------------------------------

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```
440 NEXT Z
450 PRINT @ PRINT
460 PRINT "YEARLY TOTALS:"
470 PRINT "  PAYROLL:"
480 PRINT USING 270 ; "    BASE",
    B
490 PRINT USING 270 ; "    INCENT
    IVE",I3
500 PRINT USING 270 ; "    TOTAL"
    ,T9
510 PRINT
520 PRINT "    MONTHLY REVENUE AT
    END OF"
530 PRINT USING 270 ; "CALENDAR
    YEAR",A4
540 PRINT
550 PRINT USING 270 ; "TOTAL BIL
    LINGS",A9
560 PRINT
570 PRINT "    TOTAL PAYROLL AS A
    PERCENTAGE"
580 PRINT USING 270 ; "OF SALES"
    ,T9/A9*100
590 END
```

Print annual report

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION I

(Please Type)

Program Title Financial Ratios

File Name F I N R A T

Contributor's Name

Company (if applicable)

Address

City

State/Country

Zip Code/Mail Code

Machine Size: 16K ☒ 32K ☐

Peripherals Required: none

ROMs Required: none

Number of Bytes: 1,840

Program Description: This program uses income statement and balance sheet data to compute various financial ratios.

User accepts and uses this program material AT HIS/HER OWN RISK, in reliance solely upon his/her own inspection of the program material and without reliance upon any representation or description concerning the program material.
NEITHER HP NOR THE CONTRIBUTOR MAKES ANY EXPRESS OR IMPLIED WARRANTY OF ANY KIND WITH REGARD TO THIS PROGRAM MATERIAL, INCLUDING, BUT NOT LIMITED TO THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE NEITHER HP NOR THE CONTRIBUTOR SHALL BE LIABLE FOR INCIDENTAL OR CONSEQUENTIAL DAMAGES IN CONNECTION WITH OR ARISING OUT OF THE FURNISHING, USE OR PERFORMANCE OF THIS PROGRAM MATERIAL.

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II

Sample Problem (Sketch if Desired)

Compute ratios using the following data:

Net receivables	53985080	Common stock	40569128
Net inventories	141576993	Capital & earned surplus	120721744
Net inventories prev yr	109814927	Net sales	344740452
Current assets	225925178	Cost of goods sold	229779697
Total assets	323223797	S, G & A expenses	73472649
Current liabilities	107781631	Net profit	13584388
Preferred stock	0	Dividends of preferred stock	0

SOLUTION:

LOAD "FINRAT"
RUN

Enter input data as requested.

CURRENT RATIO: 2.09621451104

ACID TEST RATIO: .78261272963

RECIEVABLES TURNOVER:
6.38584792072 TIMES

AVERAGE INVENTORY TURNOVER:
1.82807589801 TIMES

LT DEBT/TOTAL CAPITALIZATION:
.2513507241

TOTAL DEBT TO EQUITY:
1.00398043264

GROSS PROFIT MARGIN:
33.346870105 %

SELLING, GENERAL AND ADMIN.
EXPENSES TO SALES:
21.3125833962 %

NET PROFIT MARGIN:
3.94036085166 %

RATE OF RETURN ON COMMON STOCK
EQUITY: 8.42214906162 %

TURNOVER RATIO: 1.06658003836
TIMES

EARNING POWER: 4.20271022833 %

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION III

Operating Limits and Warnings

Reference(s)

Variables:

Name	Description	Length	Comments
D(1)	Net receivables		
D(2)	Net inventories		
D(3)	Net inventories from previous year end		
D(4)	Current assets		
D(5)	Total assets		
D(6)	Current liabilities		
D(7)	Preferred stock		
D(8)	Common stock		
D(9)	Capital and earned surplus		
D(10)	Net sales		
D(11)	Cost of goods sold		
D(12)	Selling, general and administrative expenses		
D(13)	Net profit		
D(14)	Dividends on preferred stock		

SERIES 80 USERS' LIBRARY

USER INSTRUCTIONS

STEP	INSTRUCTIONS
-------------	---------------------

- | | |
|---|-----------------------------------|
| 1 | LOAD "FINRAT" and press RUN. |
| 2 | Enter input data as requested. |
| 3 | Financial ratios will be printed. |

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```

10 CLEAR
20 SHORT D(14)
30 DISP "NET RECIEVABLES";
40 INPUT D(1)
50 DISP @ DISP "NET INVENTORIES
   ";
60 INPUT D(2)
70 DISP @ DISP "NET INVENTORIES
   FROM PREVIOUS YEAR-END";
80 INPUT D(3)
90 DISP @ DISP "CURRENT ASSETS"
   ;
100 INPUT D(4)
110 DISP @ DISP "TOTAL ASSETS";
120 INPUT D(5)
130 DISP @ DISP "CURRENT LIABILI
   TIES";
140 INPUT D(6)
150 DISP @ DISP "PREFERRED STOCK
   ";
160 INPUT D(7)
170 DISP @ DISP "COMMON STOCK";
180 INPUT D(8)
190 DISP @ DISP "CAPITAL AND EAR
   NED SURPLUS";
200 INPUT D(9)
210 DISP @ DISP "NET SALES";
220 INPUT D(10)
230 DISP @ DISP "COST OF GOODS S
   OLD";
240 INPUT D(11)
250 DISP @ DISP "SELLING, GEN &
   ADMIN. EXPENSES";
260 INPUT D(12)
270 DISP @ DISP "NET PROFIT";
280 INPUT D(13)
290 DISP @ DISP "DIVIDENDS ON PR
   EFERRED STOCK";
300 INPUT D(14)
310 C=D(4)/D(6)
320 PRINT "CURRENT RATIO: ";C @
   PRINT
330 A=C-D(2)/D(6)
340 PRINT "ACID TEST RATIO: ";A
   @ PRINT
350 R1=D(10)/D(1)
360 PRINT "RECIEVABLES TURNOVER:
   ",R1;" TIMES" @ PRINT
370 I1=D(11)/((D(2)+D(3))/2)
380 PRINT "AVERAGE INVENTORY TUR
   NOVER:"
390 PRINT I1;" TIMES" @ PRINT
400 D0=D(5)-D(6)-D(7)-D(8)-D(9)
410 D1=D0/(D(5)-D(6))
420 PRINT "LT DEBT/TOTAL CAPITAL
   IZATION:"
430 PRINT D1 @ PRINT
440 D2=(D0+D(6))/(D(7)+D(8)+D(9)
   )
450 PRINT "TOTAL DEBT TO EQUITY:
   ";D2 @ PRINT

```

Data entry

Print results

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```
460 M=(D(10)-D(11))/D(10)
470 PRINT "GROSS PROFIT MARGIN: ";
    M*100;"%" @ PRINT
480 S1=D(12)/D(10)
490 PRINT "SELLING, GENERAL AND
    ADMIN. EXPENSES TO SALES
    : ";100*S1;"%" @ PRINT
500 M2=D(13)/D(10)
510 PRINT "NET PROFIT MARGIN: ";
    100*M2;"%" @ PRINT
520 R2=(D(13)-D(14))/(D(8)+D(9))
530 PRINT "RATE OF RETURN ON COM
    MON STOCK EQUITY: ";100*R2;
    "%" @ PRINT
540 T=D(10)/D(5)
550 PRINT "TURNOVER RATIO: ";T;"
    TIMES" @ PRINT
560 E=D(13)/D(5)
570 PRINT "EARNING POWER: ";E*10
    0;"%" @ PRINT
580 END
```

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION I

(Please Type)

Program Title Lease Income

File Name L E A S I N

Contributor's Name

Company (if applicable)

Address

City

State/Country

Zip Code/Mail Code

Machine Size: 16K ☒ 32K ☐

Peripherals Required: none

ROMs Required: none

Number of Bytes: 3,377

Program Description: This program calculates annual lease income from (U) units at sales price (S) leased at lease rate (R) for lease period (L). It also sums the total income by year over (Y) years of lease operation.

User accepts and uses this program material AT HIS/HER OWN RISK, in reliance solely upon his/her own inspection of the program material and without reliance upon any representation or description concerning the program material.
NEITHER HP NOR THE CONTRIBUTOR MAKES ANY EXPRESS OR IMPLIED WARRANTY OF ANY KIND WITH REGARD TO THIS PROGRAM MATERIAL, INCLUDING, BUT NOT LIMITED TO THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE NEITHER HP NOR THE CONTRIBUTOR SHALL BE LIABLE FOR INCIDENTAL OR CONSEQUENTIAL DAMAGES IN CONNECTION WITH OR ARISING OUT OF THE FURNISHING, USE OR PERFORMANCE OF THIS PROGRAM MATERIAL.

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II

Sample Problem (Sketch if Desired)

Calculate lease income from the following activity: (5 year lease period)

<u>Annual Shipments</u>	<u>Ave. Sales Price (K\$)</u>	<u>Ave. Lease Rate</u>	<u>Year</u>
17	32	2.6	1
22	32	2.6	2
25	29	2.6	3
25	29	2.55	4
22	29	2.55	5
14	27.5	2.55	6
	27.5	2.5	7
	27	2.5	8
	26.5	2.5	9
	26.5	2.5	10
	26	2.5	11

SOLUTION:

LOAD "LEASIN"
RUN

The program will prompt you to input data.

ANNUAL INCOME FOR PRODUCT

YEAR	INCOME
1	169.73
2	389.38
3	579.07
4	789.79
5	985.01
6	908.82
7	709.50
8	494.10
9	286.20
10	111.30
11	0.00

SERIES 80 USERS' LIBRARY PROGRAM DESCRIPTION III

Operating Limits and Warnings Lease period in years + number of years
 manufacturing product must be less than 50.

Reference(s)

Variables:

Name	Description	Length	Comments
L	Lease period (years)		
Y	# years to manufacture		
U(*)	annual shipments		
S(*)	annual sales price		
R(*)	annual lease rate (%)		
S(*)	annual income		

SERIES 80 USERS' LIBRARY
USER INSTRUCTIONS

STEP	INSTRUCTIONS
1	LOAD "LEASIN" and press RUN.
2	Enter the number of years in the lease period.
3	Enter the number of years product will be distributed.
4	For each year the product is distributed, enter the number of products to be distributed.
5	For each year the products is in the marketplace or on lease, enter the average sales price in thousands of dollars.
6	For each year the product is in the marketplace or on lease, enter the average lease rate in % of sales per month.
7	For each year the product is in the marketplace or on lease, the annual income will be printed.

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

<pre> 10 CLEAR 20 DIM A(50),R(50),S(50),U(50), X(50) 30 DISP "ON AN ANNUAL BASIS OF SHIPMENTS OVER 'Y' YEARS, TH IS PROGRAM WILL CALCULATE LEASE"; 40 DISP "INCOME BASED ON VARIAB LE-LENGTH PRODUCT LIFE (LEAS E PERIOD)" 50 DISP @ DISP @ DISP 60 DISP "ENTER LEASE PERIOD (YE ARS)"; 70 INPUT L 80 DISP @ DISP "ENTER NUMBER OF YEARS PRODUCT ISDISTRIBUTED "; 90 INPUT Y 100 CLEAR 110 DISP @ DISP "NOW ENTER THE A NNUAL SHIPMENTS FOR EACH YE AR:" 120 FOR I=1 TO Y 130 DISP "YEAR ";I;@ INPUT U(I) 140 NEXT I 150 CLEAR 160 DISP "NOW ENTER AVG SALES PR ICE FOR EACH YEAR SYSTEMS ARE INSTALLED" 170 DISP "AND FOR ALL YEARS SYST EMS REMAINON LEASE:" @ DISP 180 FOR I=1 TO Y+L 190 DISP "YEAR ";I;@ INPUT S(I) 200 NEXT I 210 CLEAR 220 DISP "NOW ENTER AVG LEASE RA TE IN % OFSALES PRICE PER MO NTH WITHOUT" 230 DISP "SERVICE FOR EACH YEAR SYSTEMS ARE INSTALLED AND REMAIN LEASED:" 240 DISP 250 FOR I=1 TO Y+L 260 DISP "YEAR ";I;@ INPUT R(I) 270 NEXT I 280 CLEAR 290 FOR I=1 TO Y+L 300 B=I 310 IF I<Y THEN 330 320 B=Y 330 L1=I-L+1 340 IF L1>0 THEN 360 350 L1=1 360 X(I)=0 370 FOR C=L1 TO B 380 X(I)=U(C)+X(I) 390 NEXT C 400 A(I)=R(I)*S(I)*12/100*X(I) 410 NEXT I 420 PRINT " ANNUAL INCOME FOR PRODUCT" </pre>	<pre> Enter lease period Enter # years to manufacture Enter annual shipments Enter annual sales price Enter annual lease rate/mo. Begin computation </pre>
-----------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------	-----------------------------------------------------------------------------------------------------------------------------------------------------------------

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```
430 PRINT
440 PRINT "      YEAR      INCOM
    E"
450 IMAGE 6X,DD,4X,7D.DD
460 FOR I=1 TO Y+L
470 PRINT USING 450 ; I,A(I)
480 NEXT I
490 PRINT @ END
```

Print annual income

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION I

(Please Type)

Program Title Make-Buy Decision Analysis

File Name M K B U Y

Contributor's Name

Company (if applicable)

Address

City

State/Country

Zip Code/Mail Code

Machine Size: 16K ☒ 32K ☐

Peripherals Required: none

ROMs Required: none

Number of Bytes: 4,483

Program Description: This program calculates the present value of the cost saving incurred by making a product as opposed to buying it. It also prints a cash flow summary for each method for each year involved.

User accepts and uses this program material AT HIS/HER OWN RISK, in reliance solely upon his/her own inspection of the program material and without reliance upon any representation or description concerning the program material.
NEITHER HP NOR THE CONTRIBUTOR MAKES ANY EXPRESS OR IMPLIED WARRANTY OF ANY KIND WITH REGARD TO THIS PROGRAM MATERIAL, INCLUDING, BUT NOT LIMITED TO THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE NEITHER HP NOR THE CONTRIBUTOR SHALL BE LIABLE FOR INCIDENTAL OR CONSEQUENTIAL DAMAGES IN CONNECTION WITH OR ARISING OUT OF THE FURNISHING, USE OR PERFORMANCE OF THIS PROGRAM MATERIAL.

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II

Sample Problem (Sketch if Desired)

Is it cheaper to make or to buy #2 pencils given the following information:

Company name	: Hewlett Packard	Annual fixed costs:	2500
Cost to buy	: .05	Corporate tax rate:	25%
Cost to make	: .04	Local tax rate	: 55
Cost of equipment:	10000	Cost of capital	: 12%
Life of equipment:	5 years	Yearly demand	: 10000
Salvage value	: 0		

SOLUTION:

LOAD "MKBUY"
RUN

Enter the input data as requested.
Results:

THE PRESENT VALUE OF THE COST TO
MAKE IS 15208.24

THE PRESENT VALUE OF THE COST TO
BUY IS 1206.95

HEWLETT PACKARD SHOULD BUY
#2 PENCIL AT A SAVINGS OF \$
14001.2919547

*** THE FLOWS ***

YEAR 1

TO BUY:

EXPENSE	0.00
CASH FLOW	0.00

TO MAKE:

EXPENSE	0.00
CASH FLOW	-466.62

*** NET	466.62
---------	--------

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II

YEAR 2

TO BUY:		
EXPENSE	500.00	
CASH FLOW	375.00	
TO MAKE:		
EXPENSE	6600.00	
CASH FLOW	1616.67	
*** NET	-1241.67	

YEAR 3

TO BUY:		
EXPENSE	500.00	
CASH FLOW	375.00	
TO MAKE:		
EXPENSE	5786.67	
CASH FLOW	1673.33	
*** NET	-1298.33	

YEAR 4

TO BUY:		
EXPENSE	500.00	
CASH FLOW	375.00	
TO MAKE:		
EXPENSE	5010.00	
CASH FLOW	1757.50	
*** NET	-1382.50	

YEAR 5

TO BUY:		
EXPENSE	500.00	
CASH FLOW	375.00	
TO MAKE:		
EXPENSE	4270.00	
CASH FLOW	1869.17	
*** NET	-1494.17	

YEAR 6

TO BUY:		
EXPENSE	500.00	
CASH FLOW	375.00	
TO MAKE:		
EXPENSE	3566.67	
CASH FLOW	2008.33	
*** NET	-1633.33	

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION III

Operating Limits and Warnings

Reference(s)

Variables:

Name	Description	Length	Comments
D\$	Company name	25	
I\$	Product name	25	
I1	Cost to buy		
I2	Cost to make		
I3	Initial investment		
I4	Investment life		
I5	Salvage value		
I7	Annual fixed costs		
I8	Corporate tax rate		
I9	Local tax on investment (\$/1000)		
J1	Cost of capital (%)		
J2	Yearly demand for product		
A(*)	Expenses (buy)		
B(*)	Cash flows (buy)		
C(*)	Expenses (make)		
D(*)	Cash flows (make)		

SERIES 80 USERS' LIBRARY
USER INSTRUCTIONS

STEP**INSTRUCTIONS**

- 1 LOAD "MKBUY" and press RUN.
- 2 Enter the input data as prompted.
- 3 Computer will print results.
- 4 Here you have the option of changing one or more input values and recomputing the result. There are 12 options 1 through 10 permit you to re-enter an input parameter. Option 11 re-computes the result and prints.

FOR EXAMPLE: To re-do the example with a different initial capital investment, type 3 and press ENDLINE. Enter the new initial capital investment figure. When the list of options is displayed again, enter 11 and press ENDLINE. The new results will be printed.

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

<pre> 10 CLEAR 20 DIM I\$(25),D\$(25),A(25),B(25) C(25),D(25) 30 C2=0 @ C8=0 @ A1=0 @ Y1=0 40 A(1)=0 @ B(1)=0 @ C(1)=0 @ D (1)=0 50 DISP "WHAT IS THE NAME OF YO UR COMPANY"; 60 INPUT D\$ 70 DISP @ DISP "ENTER THE NAME OF THE COMPONENT UNDER CONSI DERATION"; 80 INPUT I\$ 90 FOR X=1 TO 10 @ CLEAR 100 ON X GOSUB 9100,9200,9300,94 00,9500,9600,9700,9800,9900, 9950 110 NEXT X 120 CLEAR @ D(1)=D1 @ C7=D1 130 D(I4+1)=-I5*(1-I8) 140 FOR A=2 TO I4 150 C(A)=0 @ D(A)=0 160 NEXT A 170 FOR A=2 TO I4+1 180 C6=(1+J1)^A 190 A(A)=I1*J2 200 B(A)=A(A)*(1-I8) 210 C5=B(A)/C6 220 C8=C8+C5 230 S6=I3-I5 240 C1=S6*((I4+2-A)/S4) 250 C2=C2+C1 260 C3=(I3-C2)*I9 270 C4=I2*J2+I7+C3 280 C(A)=C4+C1+C(A) 290 D(A)=C4*(1-I8)-C1*I8+D(A) 300 C5=D(A)/C6 310 C7=C7+C5 320 NEXT A 325 C7=C7+I3 330 PRINT "THE PRESENT VALUE OF THE COST TOMAKE IS ";INT(C7* 100)/100 340 PRINT 350 PRINT "THE PRESENT VALUE OF THE COST TOBUY IS ";INT(C8* 100)/100 360 PRINT 370 C9=ABS(C8-C7) 380 ON SGN(C8-C7)+2 GOTO 390,450 ,410 390 C\$="BUY" 400 GOTO 420 410 C\$="MAKE" 420 PRINT D\$;" SHOULD ";C\$ 430 PRINT I\$;" AT A SAVINGS OF \$ ";C9 440 PRINT 450 IF A1=1 THEN 650 460 PRINT </pre>	<pre> Initialize Data entry Begin computation Begin printout </pre>
-------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------	------------------------------------------------------------------------

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```

470 PRINT "      *** THE FLOWS *
    **"
480 PRINT
490 FOR A=1 TO I4+1
500 PRINT "YEAR ";A @ PRINT
510 PRINT "  TO BUY:"
520 IMAGE 7X,10A,XX,50.DD
530 PRINT USING 520 ; "EXPENSE",
    A(A)
540 PRINT USING 520 ; "CASH FLOW
    ",B(A)
550 PRINT
560 PRINT "  TO MAKE:"
570 PRINT USING 520 ; "EXPENSE",
    C(A)
580 PRINT USING 520 ; "CASH FLOW
    ",D(A)
590 PRINT
600 PRINT USING 520 ; "*** NET "
    ,B(A)-D(A)
610 PRINT
620 PRINT "  -----
    -----"
630 PRINT
640 NEXT A
650 CLEAR
660 DISP "      *** ADJUSTMENT OPTI
    ONS"
670 DISP
680 DISP "      0)  END PROGRAM"
690 DISP "      1)  COST TO BUY"
700 DISP "      2)  COST TO MANUFAC
    TURE"
710 DISP "      3)  INITIAL INVESTM
    ENT COST"
720 DISP "      4)  INVESTMENT LIFE
    (YRS)"
730 DISP "      5)  INVESTMENT SALV
    AGE VALUE"
740 DISP "      6)  ANNUAL FIXED CO
    STS"
750 DISP "      7)  CORPORATE TAX R
    ATE (%)"
760 DISP "      8)  LOCAL TAX RATE
    ($/1000)"
770 DISP "      9)  CAPITAL COST (%
    )"
780 DISP "     10)  YEARLY DEMAND E
    STIMATE"
790 DISP "     11)  RE-COMPUTE COST
    S"
800 DISP "      --> ENTER OPTION";
810 INPUT O@ CLEAR
815 IF O<0 OR O>11 THEN 650
820 ON O+1 GOSUB 840,9100,9200,9
    300,9400,9500,9600,9700,9800
    ,9900,9950,120
821 IF O#5 THEN 824
822 I5=A3
823 C(I4+1)=-I5

```

Display options list

Option selection

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```

824 IF 0#10 THEN 830
825 C2=0 @ C8=0
830 GOTO 650
840 END
9100 DISP "WHAT IS THE COST TO B
      UY ONE"
9110 DISP I$;" FOB YOUR PLANT";
9120 INPUT I1
9130 RETURN
9200 DISP "ENTER THE COST TO MAN
      UFACTURE"
9210 DISP "A ";I$
9220 DISP "IN YOUR PLANT INCLUDI
      NG DIRECT MATERIALS AND LA
      BOR EXCLUDING OVERHEAD";
9230 INPUT I2
9240 RETURN
9300 DISP "WHAT IS THE INITIAL I
      NVESTMENT (COST OF THE EXT
      RA MACHINERY)";
9310 INPUT I3
9320 RETURN
9400 DISP "WHAT IS THE LIFE OF T
      HAT INVEST-MENT IN YEARS";
9405 INPUT I4
9406 IF I4>=4 THEN 9409
9407 S3=0
9408 GOTO 9413
9409 IF I4>=8 THEN 9412
9410 S3=2/3*.07*I3
9411 GOTO 9413
9412 S3=.07*I3
9413 D1=I1-S3
9420 RETURN
9500 DISP "WHAT IS THE SALVAGE V
      ALUE OF THIS INVESTMENT"
      ;
9510 INPUT I5
9520 C(I4+1)=-I5
9530 S4=0 @ FOR S5=1 TO I4 @ S4=
      S4+S5 @ NEXT S5
9540 RETURN
9600 DISP "WHAT ARE THE ANNUAL F
      IXED COSTS (SUPERVISION AND
      MAINTENANCE) INVOLVED";
9610 INPUT I7
9620 RETURN
9700 DISP "WHAT IS YOUR CORPORAT
      E TAX RATE IN PERCENT";
9710 INPUT I8
9720 IF I8>1 THEN I8=I8/100
9730 RETURN
9800 DISP "WHAT IS THE LOCAL TAX
      RATE ON THE EXTRA INVEST
      MENT ($/1000)";
9810 INPUT I9
9820 I9=I9/1000
9830 RETURN
9900 DISP "WHAT IS YOUR COST OF
      CAPITAL IN PERCENT";

```

Entry subroutines

PROGRAM LISTING

Listing

Comments

```
9910 INPUT J1
9920 IF J1>1 THEN J1=J1/100
9930 RETURN
9950 DISP "WHAT IS YOUR ESTIMATE
      OF THE      YEARLY DEMAND FO
      R ";I$;
9960 INPUT J2
9970 RETURN
```

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION I

(Please Type)

Program Title Loan Amortization

File Name L O A N

Contributor's Name

Company (if applicable)

Address

City

State/Country

Zip Code/Mail Code

Machine Size: 16K ☒ 32K ☐

Peripherals Required: none

ROMs Required: none

Number of Bytes: 1,613

Program Description: This program amortizes a loan using arbitrary monthly payments entered by the user. Monthly reports are generated indicating payments to interest and principal. Yearly reports are printed showing accumulated payments to interest and principal and the remaining balance. Final payment is computed when balance falls below user-set maximum monthly payment.

User accepts and uses this program material AT HIS/HER OWN RISK, in reliance solely upon his/her own inspection of the program material and without reliance upon any representation or description concerning the program material.
NEITHER HP NOR THE CONTRIBUTOR MAKES ANY EXPRESS OR IMPLIED WARRANTY OF ANY KIND WITH REGARD TO THIS PROGRAM MATERIAL, INCLUDING, BUT NOT LIMITED TO THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE NEITHER HP NOR THE CONTRIBUTOR SHALL BE LIABLE FOR INCIDENTAL OR CONSEQUENTIAL DAMAGES IN CONNECTION WITH OR ARISING OUT OF THE FURNISHING, USE OR PERFORMANCE OF THIS PROGRAM MATERIAL.

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II**Sample Problem** (Sketch if Desired)

Starting month : September (9)
 Maximum Pmt./Mo: 75
 Principal : 450
 Interest rate : 10%
 Payments : 75, 50, 60, 50, 70, 75, 55

SOLUTION:

LOAD "LOAN"
 RUN

Enter input parameters as prompted.

PRINCIPAL = 450
 INT. RATE = 10 %
 MAX. PMT. = 75

MONTH 10
 PAYMENT = 75
 INTEREST = 3.75
 CUM INT = 3.75
 PRINCIPAL = 71.25
 CUM PRINC = 71.25
 BALANCE = 378.75

MONTH 11
 PAYMENT = 50
 INTEREST = 3.16
 CUM INT = 6.91
 PRINCIPAL = 46.84
 CUM PRINC = 118.09
 BALANCE = 331.91

MONTH 12
 PAYMENT = 60
 INTEREST = 2.77
 CUM INT = 9.67
 PRINCIPAL = 57.23
 CUM PRINC = 175.33
 BALANCE = 274.67

 YEAR 1
 TOTAL PAY MADE = 185
 YEAR CUM INT = 9.67
 YEAR PRINC PAY = 175.33
 BALANCE = 274.67

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II

MONTH 1
 PAYMENT = 50
 INTEREST = 2.29
 CUM INT = 11.96
 PRINCIPAL= 47.71
 CUM PRINC= 223.04
 BALANCE = 226.96

MONTH 2
 PAYMENT = 70
 INTEREST = 1.89
 CUM INT = 13.85
 PRINCIPAL= 68.11
 CUM PRINC= 291.15
 BALANCE = 158.85

MONTH 3
 PAYMENT = 75
 INTEREST = 1.32
 CUM INT = 15.18
 PRINCIPAL= 73.68
 CUM PRINC= 364.82
 BALANCE = 85.18

MONTH 4
 PAYMENT = 55
 INTEREST = .71
 CUM INT = 15.89
 PRINCIPAL= 54.29
 CUM PRINC= 419.11
 BALANCE = 30.89

MONTH 5
 PAYMENT = 31.14
 INTEREST = .26
 CUM INT = 16.14
 PRINCIPAL= 30.89
 CUM PRINC= 450
 BALANCE = 0

 YEAR 2
 TOTAL PAY MADE = 281.14
 YEAR CUM INT = 6.47
 YEAR PRINC PAY = 274.67
 BALANCE = 0

8 PAYMENTS WERE MADE

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION III

Operating Limits and Warnings

Reference(s)

Variables:

Name	Description	Length	Comments
N1	Starting month		
M	Maximum allowable payment		
P	Principal		
R	Interest rate		
X	Payment		
I	Interest		
I1	Cumulative interest		
P2	Principal		
P3	Cumulative principal		
X6	Total payments		
I6	Yearly cumulative interest		
P6	Yearly cumulative principal		
Y	Year number		

SERIES 80 USERS' LIBRARY
USER INSTRUCTIONS

STEP**INSTRUCTIONS**

- 1 LOAD "LOAN" and press RUN.
- 2 Enter principal, interest rate, maximum monthly payment as prompted.
- 3 Enter each monthly payment as prompted.
- 4 Monthly and yearly reports will be printed as payments are entered.

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

10 CLEAR	
20 DEF FNA(X) = INT((X*1000+5)/	
10)/100	
30 DISP "ENTER STARTING MONTH (
FEB=2)";	
40 INPUT N1	Data entry
50 DISP @ DISP "ENTER MAXIMUM A	
LLOWABLE PAYMENT PER MONTH";	
60 INPUT M	
70 Y=1 @ Z=1 @ I1=0 @ N=0 @ P3=	
0	Housekeeping
80 DISP @ DISP "ENTER PRINCIPAL	
";	
90 INPUT P	
100 DISP @ DISP "ENTER INTEREST	
RATE";	
110 INPUT R	
120 PRINT "PRINCIPAL =";P	
130 PRINT "INT. RATE =";R;"%"	
140 PRINT "MAX. PMT. =";M	
150 PRINT @ PRINT	
160 R=R/100	
170 P6=0 @ I6=0 @ X6=0	
180 DISP @ DISP "ENTER PAYMENT";	Enter payment
190 INPUT X	
200 IF X>=0 AND X<=M THEN 220	
210 BEEP @ DISP "Invalid payment	Total payments
. Max = ";M @ DISP @ GOTO 18	
0	Total interest
220 X6=X6+X	
230 I=P*R/12	
240 I6=I6+I	
250 I1=I1+I	
260 N1=N1+1	
270 N=N+1	
280 IF NOT Z OR X THEN 310	
290 P2=0	
300 GOTO 360	
310 P2=X-I	
320 IF P2>=0 THEN 360	
330 P=P+ABS(P2)	
340 P2=0	
350 GOTO 290	
360 P3=P3+P2	
370 P6=P6+P2	
380 IF X THEN 410	
390 P=P+I	Print report
400 GOTO 420	
410 P=P-P2	
420 PRINT "MONTH ";N1	
430 PRINT " PAYMENT = ";FNA(
X)	
440 PRINT " INTEREST = ";FNA(
I)	
450 PRINT " CUM INT = ";FNA(
I1)	
460 PRINT " PRINCIPAL= ";FNA(
P2)	
470 PRINT " CUM PRINC= ";FNA(
P3)	

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

480 PRINT " BALANCE = ";FNA(P)	
490 IF Z=0 THEN 550	
500 IF N1=12 THEN 550	Compute final payment
510 IF P>M THEN 180	
520 X=P+P*R/12	Print yearly forms
530 Z=0	
540 GOTO 220	
550 PRINT @ PRINT "----- -----"	
560 PRINT	
570 PRINT "YEAR ";Y	
580 PRINT " TOTAL PAY MADE =";F NA(X6)	
590 PRINT " YEAR CUM INT =";F NA(I6)	
600 PRINT " YEAR PRINC PAY =";F NA(P6)	
610 PRINT " BALANCE =";F NA(P)	
620 PRINT @ PRINT "----- -----"	
630 PRINT	
640 Y=Y+1 @ N1=0	
650 IF Z=0 THEN 670	
660 GOTO 170	
670 PRINT N;" PAYMENTS WERE MADE "	
680 END	

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION I

(Please Type)

Program Title Mortgage Analysis

File Name M O R G A G

Contributor's Name

Company (if applicable)

Address

City

State/Country

Zip Code/Mail Code

Machine Size: 16K ☒ 32K ☐

Peripherals Required: none

ROMs Required: none

Number of Bytes: 3,357

Program Description: MORGAG will find the missing parameter of the following four, given the remaining three: interest rate, life (years), amount borrowed, and monthly payment. The program will then print a monthly or yearly summary indicating the amount of interest, amount of payment and outstanding balance for each period.

User accepts and uses this program material AT HIS/HER OWN RISK, in reliance solely upon his/her own inspection of the program material and without reliance upon any representation or description concerning the program material.
NEITHER HP NOR THE CONTRIBUTOR MAKES ANY EXPRESS OR IMPLIED WARRANTY OF ANY KIND WITH REGARD TO THIS PROGRAM MATERIAL, INCLUDING, BUT NOT LIMITED TO THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE NEITHER HP NOR THE CONTRIBUTOR SHALL BE LIABLE FOR INCIDENTAL OR CONSEQUENTIAL DAMAGES IN CONNECTION WITH OR ARISING OUT OF THE FURNISHING, USE OR PERFORMANCE OF THIS PROGRAM MATERIAL.

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II

Sample Problem (Sketch if Desired)

Select unknown variable: Pmt (Key #4)
 Nominal annual rate : 10.5
 Mortgage life (yrs,mos): 6,0
 Borrowed amount : 10000
 Settlement date (mo,yr): 1,1980
 Table length (years) : 6
 Print annual summary : 1

SOLUTION:

LOAD "MORGAG".
 RUN

Enter input parameters as prompted.

MORTGAGE TERMS

NOMINAL ANNUAL RATE = 10.5 %
 MORTGAGE LIFE = 6 YRS 0 MONTHS
 AMOUNT BORROWED = 10000
 MONTHLY PAYMENT = 187.79

MORTGAGE TABLE

	INTEREST	PRINCIPAL	ENDING BALANCE
1980	912.93	1152.76	8847.24
1981	863.32	1390.16	7457.09
1982	710.09	1543.39	5913.70
1983	540.01	1713.47	4200.23
1984	351.20	1902.28	2297.96
1985	141.56	2111.92	186.04

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION III

Operating Limits and Warnings

Reference(s)

Variables:

Name	Description	Length	Comments
R	Interest rate		
Y	Years of mortgage life		
M	Months of mortgage life		
A	Amount borrowed		
P	Monthly payment		
T1	Month of settlement		
T2	Year of settlement		
T3	Table length (years)		
Z1	Summary selection		
I1, S1	Monthly, Yearly interest PAID		
P1, S2	Monthly, Yearly principal PAID		
A1	Monthly starting balance		
A2	Unpaid principal		
T2	Current year for table		
M2	Current month for table		

SERIES 80 USERS' LIBRARY
USER INSTRUCTIONS

STEP**INSTRUCTIONS**

- 1 LOAD "MORGAG" and press RUN.
- 2 Select Special Function key corresponding to unknown variable.
- 3 Enter remaining three variables as prompted.
- 4 Enter settlement date, table length, and summary selections as prompted.
- 5 Tables will be printed.

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```

10 DIM M$(36)
20 CLEAR
25 DEF FNA(X) = INT((X*1000+5)/
10)/100
30 M$="JANFEBMARAPRMAYJUNJULAug
SEP OCT NOV DEC"
40 ON KEY# 1,"RATE" GOTO 110
50 ON KEY# 2,"LIFE" GOTO 120
60 ON KEY# 3,"AMOUNT" GOTO 130
70 ON KEY# 4,"PMT" GOTO 140
80 KEY LABEL @ DISP
90 DISP "SELECT UNKNOWN VARIABLE:"
100 GOTO 100
110 Z=1 @ GOTO 150
120 Z=2 @ GOTO 150
130 Z=3 @ GOTO 150
140 Z=4
150 CLEAR
160 !
170 !
180 IF Z=1 THEN 240
190 DISP "ENTER NOMINAL ANNUAL R
ATE (%)" ;
200 INPUT R
210 R=R/100
220 DISP
230 IF Z=2 THEN 280
240 DISP "ENTER MORTGAGE LIFE (Y
RS, MONTHS)" ;
250 INPUT Y,M
260 DISP
270 IF Z=3 THEN 320
280 DISP "ENTER AMOUNT TO BE BOR
ROWED" ;
290 INPUT A
300 DISP
310 IF Z=4 THEN 350
320 DISP "ENTER MONTHLY PAYMENT"
;
330 INPUT P
340 DISP
350 DISP "ENTER SETTLEMENT DATE
(MO, YEAR)" ;
360 INPUT T1,T2
370 DISP
380 DISP "ENTER TABLE LENGTH (YE
ARS)" ;
390 INPUT T3
400 DISP @ DISP "PRINT ANNUAL (1
) OR MONTHLY (2) SUMMARY" ;
410 INPUT Z1
420 CLEAR
430 IF Z=2 THEN 460
440 N=12*Y+M
450 IF Z=1 THEN 620
460 R=R/12
470 IF Z=3 THEN 570
480 IF Z=4 THEN 600
490 IF A*R/P<1 THEN 520

```

Set up option selection

Wait for special function key

Data entry

Error trap

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

<pre> 500 BEEP @ DISP "THE FIRST MONTH S PAYMENT WILL NOT COVER I T'S INTEREST" 510 STOP 520 N=-LOG(1-A*R/P)/LOG(1+R) 530 N=INT(N)+1 540 Y=INT(N/12) 550 M=N-12*Y 560 GOTO 740 570 A=P*(1-1/(1+R)^N)/R 580 A=INT((A*100+5)/10)/10 590 GOTO 740 600 P=A*R*(1+R)^N/((1+R)^N-1) 610 GOTO 740 620 R=(P*N/A-1)/12 630 R1=R 640 IF R>0 THEN 690 650 BEEP @ DISP "RATE IS NEGATIV E OR ZERO" 660 STOP 670 R=R-2*R1 680 R=R+R1 690 C=P*(1/(R/((1+R)^N-1)+R)) 700 IF ABS(C-A)<.01 THEN 740 710 R1=R1/2 720 IF C-A<0 THEN 670 730 GOTO 680 740 IF 12*P>R*A*12+1 THEN 800 750 BEEP 760 DISP "YOUR FIRST YEARS PAYME NTS ARE ";12*P 770 DISP @ DISP "THE FIRST YEAR' S INTEREST IS ";R*A*12 780 DISP @ DISP "THEREFORE, THE LIFE OF THE MORTGAGE IS UNDEFINED" 790 STOP 800 ! 810 PRINT " MORTGAGE TE RMS" 820 PRINT 830 PRINT "NOMINAL ANNUAL RATE = ";FNA(R*1200);"% " 840 PRINT 850 PRINT "MORTGAGE LIFE =";Y;"Y RS";M;"MONTHS" 860 PRINT 870 PRINT "AMOUNT BORROWED =";A 880 PRINT 890 PRINT "MONTHLY PAYMENT =";FN A(P) 900 PRINT 910 IF Z#2 THEN 930 920 PRINT "MORTGAGE LIFE HAS BEE N ROUNDEDUP TO THE NEAREST MONTH" @ PRINT 930 PRINT "----- -----" 940 PRINT 950 PRINT " MORTGAGE TAB LE" </pre>	<pre> Compute life Compute amount borrowed Compute payment Compute rate Error trap Print terms </pre>
-------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------	-------------------------------------------------------------------------------------------------------

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```

960 PRINT
970 Z2=0 @ S1=0 @ S2=0
980 IF T1=12 THEN 1000
990 M2=T1 @ GOTO 1020
1000 T2=T2+1
1010 M2=0
1020 M3=M2+1
1030 IF Z1=2 THEN PRINT "  FOR T
      HE CALENDER YEAR ";T2 @ PRI
      NT
1040 IF Z1=1 THEN PRINT "
      ENDING"
1050 IF Z1=2 THEN PRINT "
      STARTING"
1060 PRINT "      INTEREST PRINC
      IPAL BALANCE"
1070 FOR M1=M3 TO 12*T3
1080 I1=FNA(A*R)
1090 IF P<A+I1 THEN P1=P-I1 ELSE
      P1=A
1100 A1=A @ A=A1-P1
1110 S1=S1+I1 @ S2=S2+P1
1120 M2=M2+1
1130 IF Z1=1 THEN 1320
1140 IMAGE 4A,50.DD,2(XX,50.DD)
1150 PRINT USING 1140 ; M$(3*M2-
      2,3*M2),I1,P1,A1
1160 IF M2=12 THEN 1190
1170 IF A>0 THEN 1400
1180 Z2=1
1190 PRINT
1200 PRINT "PRINCIPAL REPAID =";
      FNA(S2)
1210 PRINT "INTEREST PAID      =";
      FNA(S1)
1220 PRINT "UNPAID PRINCIPAL =";
      FNA(A)
1230 PRINT
1240 IF Z2=1 THEN 1420
1250 T2=T2+1
1260 PRINT "-----"
      "
1270 PRINT
1280 IF M1=12*T3 THEN 1410
1290 PRINT "  FOR THE CALENDAR Y
      EAR";T2
1300 PRINT
1310 GOTO 1390
1320 IF M2=12 THEN 1350
1330 IF A>0 THEN 1400
1340 Z2=1
1350 PRINT USING 1140 ; VAL$(T2)
      ,S1,S2,A
1360 T2=T2+1
1370 IF M1=12*T3 THEN 1410
1380 IF Z2=1 THEN 1410
1390 S1=0 @ S2=0 @ M2=0
1400 NEXT M1
1410 PRINT
1420 END

```

Print monthly report

Print yearly summary

Print yearly report

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION I

(Please Type)

Program Title Cost of Equity Capital

File Name E Q U I T Y

Contributor's Name

Company (if applicable)

Address

City

State/Country

Zip Code/Mail Code

Machine Size: 16K ☒ 32K ☐

Peripherals Required: none

ROMs Required: none

Number of Bytes: 3,040

Program Description: This program will compute the cost of equity capital by computing dividends and the share price for future periods. It will then find the discount rate by equating the present stream value to the current share price.

User accepts and uses this program material AT HIS/HER OWN RISK, in reliance solely upon his/her own inspection of the program material and without reliance upon any representation or description concerning the program material.
NEITHER HP NOR THE CONTRIBUTOR MAKES ANY EXPRESS OR IMPLIED WARRANTY OF ANY KIND WITH REGARD TO THIS PROGRAM MATERIAL, INCLUDING, BUT NOT LIMITED TO THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE NEITHER HP NOR THE CONTRIBUTOR SHALL BE LIABLE FOR INCIDENTAL OR CONSEQUENTIAL DAMAGES IN CONNECTION WITH OR ARISING OUT OF THE FURNISHING, USE OR PERFORMANCE OF THIS PROGRAM MATERIAL.

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II

Sample Problem (Sketch if Desired)

Compute cost of equity capital with the following conditions:

Current Price/Share = \$100

Dividends/Share = \$10

<u>Segment #</u>	<u>Growth Rate</u>	<u>Last Effective Period</u>
1	.02	2
2	.03	4
3	.04	6
4	.05	10
5	.06	12
6	.07	15
7	.08	20
8	.10	25

SOLUTION:

LOAD "EQUITY"

RUN

Enter input parameters as prompted

```

SHARE PRICE:      100
DIVIDEND:         10
INITIAL GROWTH RATE: 2 %
  
```

```

COST OF EQUITY CAPITAL= 15.74 %
-----
  
```

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION III

Operating Limits and Warnings Enter growth rate as decimal less than one.

Reference(s) Computations are based on the Gordon model.

Variables:

Name	Description	Length	Comments
P	Current price/share		
D(*)	Dividends/share		
N	# of growth segments		
R	Cost of equity capital		
M	# of iterations		

SERIES 80 USERS' LIBRARY
USER INSTRUCTIONS

STEP**INSTRUCTIONS**

- 1 LOAD "EQUITY" and press RUN.
- 2 Enter current price per share and dividends per share.
- 3 Enter the number of growth segments
- 4 For each growth segment, enter the growth rate as a decimal less than one, and the last effective period for the segment.
- 5 Results will be printed.
- 6 You now have four options:
 - a) NEW RUN: go to Step 2 (enter new problem).
 - b) GROWTH : go to Step 3 (enter new growth information).
 - c) PRICE : go to Step 2 (enter new prices), then go to Step 5.
 - d) END : stops program.

Select option by pressing appropriate special function key.

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```

10 DIM D(100),L(20),B(20),G(20)
20 Q=0
30 CLEAR
40 DISP "ENTER THE CURRENT PRIC
E/SHARE";
50 INPUT P
60 DISP @ DISP "ENTER CURRENT D
IVIDENDS/SHARE";
70 INPUT D(1)
80 IF Q=2 THEN 180
90 DISP @ DISP "ENTER THE NUMBE
R OF GROWTH          SEGMENTS";
100 INPUT N
110 CLEAR @ DISP "FOR EACH GROWT
H SEGMENT, ENTER THE GROWTH
RATE IN DECIMAL AND"
120 DISP "THE LAST EFFECTIVE PER
IOD FOR  EACH SEGMENT" @ DI
SP
130 FOR I=1 TO N
140 DISP "SEGMENT ";I;
150 INPUT G(I),L(I)
160 NEXT I
170 CLEAR
180 R=0
190 B(1)=1
200 IF N=1 THEN R=100*(D(1)/P+G(
1)) @ GOTO 290
210 FOR I=1 TO N-1
220 FOR J=B(I) TO L(I)
230 D(J+1)=D(J)*(1+G(I))
240 NEXT J
250 B(I+1)=L(I)+1
260 R=R+G(I)
270 NEXT I
280 GOSUB 450
290 PRINT "SHARE PRICE:";TAB(20)
;P
300 PRINT "DIVIDEND:";TAB(20);D(
1)
310 PRINT "INITIAL GROWTH RATE:"
;INT(G(1)*100);"%"
320 PRINT @ PRINT "COST OF EQUIT
Y CAPITAL=";INT(R*100)/100;"
%"
330 PRINT "-----"
-----"
340 PRINT
350 ON KEY# 1,"NEW RUN" GOTO 410
360 ON KEY# 2," GROWTH" GOTO 420
370 ON KEY# 3," PRICE" GOTO 430
380 ON KEY# 4," END" GOTO 440
390 CLEAR @ KEY LABEL @ DISP "
SELECT OPTION"
400 GOTO 400
410 CLEAR @ GOTO 20
420 Q=1 @ CLEAR @ GOTO 90
430 Q=2 @ CLEAR @ GOTO 40
440 CLEAR @ END
450 R=(R+G(N))/N+D(1)/P

```

Data entry

Case of one growth period

Computation

Print results

Re-run

Option selection

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```

460 M=0
470 R=R-.55
480 D1=1
490 FOR I=1 TO 13
500 R=R+D1
510 M=M+1
520 D1=D1/2
530 R=R-D1
540 F=1
550 IF M>100 THEN 680
560 IF R=-1 THEN 650
570 Z=-P
580 FOR J=1 TO L(N-1)
590 F=F/(1+R)
600 Z=Z+D(J)*F @ NEXT J
610 IF R<G(N) THEN 500
620 Z=Z+D(L(N-1))*F/(R-G(N))
630 IF Z<0 THEN 530
640 NEXT I
650 R=100*R
660 R=INT(100*R+.5)/100
670 RETURN
680 PRINT
690 PRINT "TOO MANY ITERATIONS."
700 PRINT "LAST COST WAS ";R*100
    ;"PERCENT"
710 PRINT "WITH VALUE $";Z
720 GOTO 350

```

Computation

Error trap

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION I

(Please Type)

Program Title Depreciation Method Comparison

File Name D E P C O M

Contributor's Name

Company (if applicable)

Address

City

State/Country

Zip Code/Mail Code

Machine Size: 16K ☒ 32K ☐

Peripherals Required: none

ROMs Required: none

Number of Bytes: 7,764

Program Description: This program computes and prints the monthly depreciation of a given investment by four methods: straight line, double declining balance, sum of the years digits, and 150% declining balance.

User accepts and uses this program material AT HIS/HER OWN RISK, in reliance solely upon his/her own inspection of the program material and without reliance upon any representation or description concerning the program material.
NEITHER HP NOR THE CONTRIBUTOR MAKES ANY EXPRESS OR IMPLIED WARRANTY OF ANY KIND WITH REGARD TO THIS PROGRAM MATERIAL, INCLUDING, BUT NOT LIMITED TO THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE NEITHER HP NOR THE CONTRIBUTOR SHALL BE LIABLE FOR INCIDENTAL OR CONSEQUENTIAL DAMAGES IN CONNECTION WITH OR ARISING OUT OF THE FURNISHING, USE OR PERFORMANCE OF THIS PROGRAM MATERIAL.

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II

Sample Problem (Sketch if Desired)

Amount of investment : 35000
 Salvage value : 15000
 Depreciable life : 5 years
 Investment made in : 07,1969
 Discount rate : .15
 Automatic switchover : 1
 Yearly summary desired: Y

SOLUTION:

LOAD "DEPCOM"
 RUN

Enter input parameters as prompted

TOTAL 1969
 STRAIGHT LINE = 1666.67
 DOUBLE DECLINING = 3333.33
 SUM OF YEARS DIGITS= 2777.78
 150% DECLINING = 2500.00

CUMULATIVE DEPRECIATION:

 STRAIGHT LINE = 1666.67
 DOUBLE DECLINING = 3333.33
 SUM OF YEARS DIGITS= 2777.78
 150% DECLINING = 2500.00

UNDEPRECIATED BALANCE:

 STRAIGHT LINE = 18333.33
 DOUBLE DECLINING = 16666.67
 SUM OF YEARS DIGITS= 17222.22
 150% DECLINING = 17500.00

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II

TOTAL 1970
 STRAIGHT LINE = 4000.00
 DOUBLE DECLINING = 6666.67
 SUM OF YEARS DIGITS= 6111.11
 150% DECLINING = 5250.00

CUMULATIVE DEPRECIATION:

STRAIGHT LINE = 5666.67
 DOUBLE DECLINING = 10000.00
 SUM OF YEARS DIGITS= 8888.89
 150% DECLINING = 7750.00

UNDEPRECIATED BALANCE:

STRAIGHT LINE = 14333.33
 DOUBLE DECLINING = 10000.00
 SUM OF YEARS DIGITS= 11111.11
 150% DECLINING = 12250.00

TOTAL 1971
 STRAIGHT LINE = 4000.00
 DOUBLE DECLINING = 3800.00
 SUM OF YEARS DIGITS= 4777.78
 150% DECLINING = 3675.00

CUMULATIVE DEPRECIATION:

STRAIGHT LINE = 9666.67
 DOUBLE DECLINING = 13800.00
 SUM OF YEARS DIGITS= 13666.67
 150% DECLINING = 11425.00

UNDEPRECIATED BALANCE:

STRAIGHT LINE = 10333.33
 DOUBLE DECLINING = 6200.00
 SUM OF YEARS DIGITS= 6333.33
 150% DECLINING = 8575.00

TOTAL 1972
 STRAIGHT LINE = 4000.00
 DOUBLE DECLINING = 2400.00
 SUM OF YEARS DIGITS= 3444.44
 150% DECLINING = 2572.50

CUMULATIVE DEPRECIATION:

STRAIGHT LINE = 13666.67
 DOUBLE DECLINING = 16200.00
 SUM OF YEARS DIGITS= 17111.11
 150% DECLINING = 13997.50

UNDEPRECIATED BALANCE:

STRAIGHT LINE = 6333.33
 DOUBLE DECLINING = 3800.00
 SUM OF YEARS DIGITS= 2888.89
 150% DECLINING = 6002.50

TOTAL 1973
 STRAIGHT LINE = 4000.00
 DOUBLE DECLINING = 2400.00
 SUM OF YEARS DIGITS= 2111.11
 150% DECLINING = 1800.75

CUMULATIVE DEPRECIATION:

STRAIGHT LINE = 17666.67
 DOUBLE DECLINING = 18600.00
 SUM OF YEARS DIGITS= 19222.22
 150% DECLINING = 15798.25

UNDEPRECIATED BALANCE:

STRAIGHT LINE = 2333.33
 DOUBLE DECLINING = 1400.00
 SUM OF YEARS DIGITS= 777.78
 150% DECLINING = 4201.75

TOTAL 1974
 STRAIGHT LINE = 2333.33
 DOUBLE DECLINING = 1400.00
 SUM OF YEARS DIGITS= 777.78
 150% DECLINING = 840.35

CUMULATIVE DEPRECIATION:

STRAIGHT LINE = 20000.00
 DOUBLE DECLINING = 20000.00
 SUM OF YEARS DIGITS= 20000.00
 150% DECLINING = 16638.60

UNDEPRECIATED BALANCE:

STRAIGHT LINE = 0.00
 DOUBLE DECLINING = 0.00
 SUM OF YEARS DIGITS= 0.00
 150% DECLINING = 3361.40

THE PRESENT VALUE OF THE
 DEPRECIATION AT THE START OF
 1969 AT .15 IS AS FOLLOWS:

STRAIGHT LINE = 13408.62
 DOUBLE DECLINING = 14729.48
 SUM OF YEARS DIGITS= 14647.51
 150% DECLINING = 12219.19

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION III

Operating Limits and Warnings Depreciable life must be greater than two years and less than 51 years.

Reference(s)

Variables:

Name	Description	Length	Comments
I1	Amount of investment		
S1	Salvage value		
L1	Depreciation life (years)		
A1	Month of investment date		
Z1	Year of investment date		
R	Discount rate		
Y	Switchover option		
W(I)	Straightline depreciation		
X(I)	Double declining depreciation		
Y(I)	Sum of years digits depreciation		
Z(I)	150% declining depreciation		

SERIES 80 USERS' LIBRARY
USER INSTRUCTIONS

STEP**INSTRUCTIONS**

- 1 LOAD "DEPCOM" and press RUN.
- 2 Enter input parameters as prompted.
- 3 Enter switchover option:
 - 0) Prevent switchover.
 - 1) Automatic switchover from double declining balance method to straightline depreciation when annual straightline depreciation becomes greater than the double declining balance value.
 - Year #) Specifies in which year switchover should take place.
- 4 If annual summary desired only enter 'Y', else enter 'N' to get monthly and annual summary.
- 5 Results will be printed.

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```

10 CLEAR
20 DIM A(50),B(50),C(50),D(50)
30 DIM H(50),I(50),J(50),K(50)
40 DIM W(12),X(12),Y(12),Z(12)
50 DISP "ENTER THE AMOUNT OF IN
  VESTMENT";
60 INPUT I1
70 DISP @ DISP "ENTER THE SALVA
  GE VALUE";
80 INPUT S1
90 DISP @ DISP "WHAT IS THE DEP
  RECIABLE LIFE";
100 INPUT L1
110 IF L1>2 AND L1<51 THEN 150
120 BEEP @ DISP "Life too ";
130 IF L1<3 THEN DISP "short" @
  GOTO 90
140 DISP "long" @ GOTO 90
150 DISP @ DISP "ENTER INVESTMEN
  T DATE (MM,19YY)";
160 INPUT A1,Z1
170 DISP @ DISP "ENTER THE DISCO
  UNT RATE (N<1) FOR COMPUTI
  NG THE PRESENT VALUE"
180 DISP "OF THE ANNUAL DEPRECIA
  TION";
190 INPUT R@ CLEAR
200 DISP @ DISP "          SWITCHOVE
  R OPTIONS:"
210 DISP @ DISP "          0) NO
  SWITCHOVER"
220 DISP "          1) AUTOMATIC"
230 DISP "          Year#) SPECIFIC T
  IME"
240 DISP @ DISP "ENTER OPTION:";
250 INPUT Y
260 CLEAR @ DISP "DO YOU WISH A
  YEARLY SUMMARY ONLY";
270 INPUT Z#@ CLEAR
280 K=0 @ P=0 @ S=0
290 Q1=0 @ Q2=0 @ Q3=0 @ Q4=0
300 V1=0 @ V2=0 @ V3=0 @ V4=0
310 ! STRAIGHTLINE
320 N=S1
330 FOR I=1 TO L1
340 A(I)=(I1-S1)/L1
350 H(I)=A(I)/12
360 NEXT I
370 ! DOUBLE DECLINING BALANCE
380 N=S1
390 FOR I=1 TO L1
400 B(I)=2*((I1-N)/L1)
410 I(I)=B(I)/12
420 M=N
430 N=N+B(I)
440 GOTO 470
450 NEXT I
460 GOTO 700
470 IF Y<1 THEN 450
480 IF Y>1 THEN 600

```

Data entry

Housekeeping

Straight line

Double declining balance

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```

490 IF I=1 THEN 450
500 IF P=1 THEN 570
510 IF B(I)>A(I) THEN 450
520 G=L1+1-I
530 P=1
540 B(I)=(I1-M)/G
550 I(I)=B(I)/12
560 GOTO 450
570 B(I)=B(I-1)
580 I(I)=B(I)/12
590 GOTO 450
600 G=Y+1-Z1
610 IF I<G THEN 450
620 IF P=1 THEN 670
630 P=1
640 B(I)=(I1-M)/(L1+1-G)
650 I(I)=B(I)/12
660 GOTO 450
670 B(I)=B(I-1)
680 I(I)=B(I)/12
690 GOTO 450
700 ! SUM OF YEARS DIGITS
710 N=S1
720 FOR I=1 TO L1
730 C(I)=(I1-N)*(L1+1-I)*(2/(L1*
    (L1+1)))
740 J(I)=C(I)/12
750 NEXT I
760 ! 150% DECLINING BALANCE
770 N=S1
780 FOR I=1 TO L1
790 D(I)=1.5*((I1-N)/L1)
800 K(I)=D(I)/12
810 N=N+D(I)
820 NEXT I
830 I=0
840 K=K+1
850 X=K+(Z1-1)
860 IF K<=L1+1 THEN 1040
870 FOR I=1 TO L1
880 V1=V1+A(I)/(1+R)^I
890 V2=V2+B(I)/(1+R)^I
900 V3=V3+C(I)/(1+R)^I
910 V4=V4+D(I)/(1+R)^I
920 NEXT I
930 PRINT
940 PRINT "THE PRESENT VALUE OF
    THE"
950 PRINT "DEPRECIATION AT THE S
    TART OF"
960 PRINT Z1;" AT";R;" IS AS FOL
    LOWS:"
970 PRINT
980 PRINT USING 1740 ; V1
990 PRINT USING 1750 ; V2
1000 PRINT USING 1760 ; V3
1010 PRINT USING 1770 ; V4
1020 PRINT
1030 END
1040 I=I+1

```

Sum of the years digits

150% declining balance

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```

1050 IF I>12 THEN 1310
1060 IF I<=A1 THEN 1160
1070 S=1
1080 IF K<L1+1 THEN 1110
1090 W(I)=0 @ X(I)=0 @ Y(I)=0 @
      Z(I)=0
1100 GOTO 1230
1110 W(I)=H(K)
1120 X(I)=I(K)
1130 Y(I)=J(K)
1140 Z(I)=K(K)
1150 GOTO 1230
1160 IF S=0 THEN 1220
1170 W(I)=H(K-1)
1180 X(I)=I(K-1)
1190 Y(I)=J(K-1)
1200 Z(I)=K(K-1)
1210 GOTO 1230
1220 W(I)=0 @ X(I)=0 @ Y(I)=0 @
      Z(I)=0
1230 ! PRINTOUT ROUTINE
1240 IF Z$(1,1)="Y" THEN 1040
1250 PRINT VAL$(I);"/";VAL$(X-19
      00)
1260 PRINT USING 1740 ; W(I)
1270 PRINT USING 1750 ; X(I)
1280 PRINT USING 1760 ; Y(I)
1290 PRINT USING 1770 ; Z(I)
1300 PRINT @ GOTO 1040
1310 ! TOTALS ROUTINE
1320 S=1
1330 T1=0 @ T2=0 @ T3=0 @ T4=0
1340 FOR J=1 TO 12
1350 T1=T1+W(J)
1360 T2=T2+X(J)
1370 T3=T3+Y(J)
1380 T4=T4+Z(J)
1390 NEXT J
1400 Q1=Q1+T1
1410 Q2=Q2+T2
1420 Q3=Q3+T3
1430 Q4=Q4+T4
1440 PRINT "TOTAL ";X
1450 PRINT USING 1740 ; T1
1460 PRINT USING 1750 ; T2
1470 PRINT USING 1760 ; T3
1480 PRINT USING 1770 ; T4
1490 PRINT
1500 PRINT "CUMULATIVE DEPRECIAT
      ION:"
1510 PRINT
1520 PRINT USING 1740 ; Q1
1530 PRINT USING 1750 ; Q2
1540 PRINT USING 1760 ; Q3
1550 PRINT USING 1770 ; Q4
1560 PRINT
1570 B1=I1-S1-Q1
1580 IF B1<=1 THEN B1=0
1590 B2=I1-S1-Q2
1600 IF B2<=1 THEN B2=0

```

Print out routines

Totals routine

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```
1610 B3=I1-S1-Q3
1620 IF B3<=1 THEN B3=0
1630 B4=I1-S1-Q4
1640 IF B4<=1 THEN B4=0
1650 PRINT "UNDEPRECIATED BALANC
E:"
1660 PRINT
1670 PRINT USING 1740 ; B1
1680 PRINT USING 1750 ; B2
1690 PRINT USING 1760 ; B3
1700 PRINT USING 1770 ; B4
1710 PRINT
1720 PRINT " -----"
1730 GOTO 830
1740 IMAGE 3X,"STRAIGHT LINE
      = ",50.00
1750 IMAGE 3X,"DOUBLE DECLINING
      = ",50.00
1760 IMAGE 3X,"SUM OF YEARS DIGI
      TS= ",50.00
1770 IMAGE 3X,"150% DECLINING
      = ",50.00
1780 END
```

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION I

(Please Type)

Program Title Simple Loan Analysis

File Name L E N D E R

Contributor's Name

Company *(if applicable)*

Address

City

State/Country

Zip Code/Mail Code

Machine Size: 16K ☒ 32K ☐

Peripherals Required: none

ROMs Required: none

Number of Bytes: 1,313

Program Description: LENDER calculates the monthly interest charges and outstanding balance of a loan that must be paid off in one year or less.

User accepts and uses this program material AT HIS/HER OWN RISK, in reliance solely upon his/her own inspection of the program material and without reliance upon any representation or description concerning the program material.
NEITHER HP NOR THE CONTRIBUTOR MAKES ANY EXPRESS OR IMPLIED WARRANTY OF ANY KIND WITH REGARD TO THIS PROGRAM MATERIAL, INCLUDING, BUT NOT LIMITED TO THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE NEITHER HP NOR THE CONTRIBUTOR SHALL BE LIABLE FOR INCIDENTAL OR CONSEQUENTIAL DAMAGES IN CONNECTION WITH OR ARISING OUT OF THE FURNISHING, USE OR PERFORMANCE OF THIS PROGRAM MATERIAL.

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION II

Sample Problem (Sketch if Desired)

Calculate interest charges on the following loan:

Amount of loan : \$350
 Monthly payment: 95
 Interest rate : 10.5%

SOLUTION:

LOAD "LENDER"
 RUN

Enter input data as requested

AMOUNT OF LOAN: \$ 350
 MONTHLY PAYMENT: \$ 95
 INTEREST RATE: 10.5 %

 MONTH: 1
 INTEREST: 3.06
 PRINCIPAL: 91.94
 BALANCE DUE: 258.06

 MONTH: 2
 INTEREST: 2.26
 PRINCIPAL: 92.74
 BALANCE DUE: 165.32

 MONTH: 3
 INTEREST: 1.45
 PRINCIPAL: 93.55
 BALANCE DUE: 71.77

 MONTH: 4
 INTEREST: .63
 PRINCIPAL: 71.77
 BALANCE DUE: 0

 MONTHS TO PAY OFF LOAN: 4

TOTAL INTEREST CHARGES: 7.4

SERIES 80 USERS' LIBRARY

PROGRAM DESCRIPTION III

Operating Limits and Warnings Term of loan must be one year or less

Reference(s)

Variables:

Name	Description	Length	Comments
B	Amount of loan		
M	Monthly payment		
R	Interest rate		
N	Months to pay off loan		
T	Total interest charges		
I	Monthly interest		

SERIES 80 USERS' LIBRARY
USER INSTRUCTIONS

STEP**INSTRUCTIONS**

- 1 LOAD "LENDER" and press RUN.
- 2 Enter amount of loan.
- 3 Enter the monthly payment.
- 4 Enter the interest as greater than one (for example: 11.5).
- 5 Results will be printed.

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```

10 CLEAR
20 DISP "ENTER THE AMOUNT OF TH
   E LOAN";
30 INPUT B
40 DISP @ DISP "ENTER THE MONTH
   LY PAYMENT";
50 INPUT M
60 DISP @ DISP "ENTER THE INTER
   EST RATE";
70 INPUT R
80 IF R>1 THEN 110
90 BEEP @ DISP "Enter rate as p
   ercent > 1"
100 GOTO 60
110 CLEAR
120 Z=B*R/1200
130 IF (Z+B)/12<M THEN 160
140 BEEP @ DISP "LOAN CANNOT BE
   PAID OFF IN LESS THAN ONE YE
   AR. INCREASE YOUR MONTHLY
   PAYMENT"
150 DISP @ GOTO 20
160 PRINT "AMOUNT OF LOAN: $";B
170 PRINT "MONTHLY PAYMENT: $";M
180 PRINT "INTEREST RATE: ";R
   ;"%"
190 PRINT
200 PRINT "-----"
   -----" @ PRINT
210 T=0
220 X=0
230 FOR N=1 TO 37
240 I=B*R/1200
250 I=INT(100*I+.5)/100
260 B=B+I-M
270 T=T+I
280 X=X+1
290 PRINT "MONTH:";TAB(18);X
300 PRINT " INTEREST:";TAB(18);
   I
310 PRINT " PRINCIPAL:";TAB(18)
   ;M-I
320 PRINT " BALANCE DUE:";TAB(1
   8);B
330 PRINT @ PRINT "-----"
   -----"
340 PRINT
350 IF B<=M THEN 370
360 NEXT N
370 N=N+1
380 F=B
390 I=B*R/1200
400 I=INT(100*I+.5)/100
410 T=T+I
420 X=X+1 @ B=0
430 PRINT "MONTH:";TAB(18);X
440 PRINT " INTEREST:";TAB(18);
   I
450 PRINT " PRINCIPAL:";TAB(18)
   ;F

```

Data entry

Error trap

Print input data

Final payment

SERIES 80 USERS' LIBRARY

PROGRAM LISTING

Listing

Comments

```
460 PRINT "  BALANCE DUE:";TAB(1
    8);B
470 PRINT @ PRINT "-----
    -----"
480 PRINT
490 PRINT "MONTHS TO PAY OFF LOA
    N:";N
500 PRINT @ PRINT "TOTAL INTERES
    T CHARGES:";T
510 END
```

Final report

BUDGETING AND FINANCE I

PROFORMA INCOME STATEMENT AND BALANCE SHEET
SALES COMMISSION REPORT
FINANCIAL RATIOS
LEASE INCOME
MAKE—BUY DECISION ANALYSIS
LOAN AMORTIZATION
MORTGAGE ANALYSIS
COST OF EQUITY CAPITAL
DEPRECIATION METHOD COMPARISON
SIMPLE LOAN ANALYSIS



**HEWLETT
PACKARD**

For additional information please contact your local Hewlett-Packard Sales Office.